Solution to Exercise 2.6.5

If (G_1, G_2) is a Brauer pair then by Lemma 2.6.3 and Definition 2.6.1

(a) $G_1/G_1' \cong G_2/G_2'$,

(b) $\mathbf{Z}(G_1) \cong \mathbf{Z}(G_2)$,

(c) $|\{g^{G_1} \mid g \in G_1,\ |\langle g\rangle| = d\}| = |\{g^{G_2} \mid g \in G_2,\ |\langle g\rangle| = d\}|$ for all $d \mid |G_1|$.

Here (a) and (b) can be checked quite fast. Hence it is easily seen that the following GAP-function all Brauer pairs (G_1, G_2) where $G_1, G_2 \in$ gg, where gg is a collection of groups.

```
braupairli := function( gg )
local ggf, types, hh, hhf, kk, i, j, brps, tp;
brps := [];
types := Set( List( gg, g ->
         [ IdGroup(g/DerivedSubgroup(g)), IdGroup(Centre(g)) ] ));
ggf := List( types, t -> Filtered( gg, g ->
IdGroup( g/DerivedSubgroup(g) ) = t[1] and IdGroup(Centre(g)) = t[2]) );
for hh in ggf do
 types := Set( List( hh, g-> Collected(List( ConjugacyClasses(g),
               c -> Order(Representative(c)))) ) );
 hhf := List( types, t -> Filtered( hh, g ->
 Collected(List(ConjugacyClasses(g), c -> Order(Representative(c)))) = t) );
 for kk in  hhf do
 for i in [1..Length(kk)] do
  for j in [i+1..Length(kk)] do
   tp := TransformingPermutationsCharacterTables
                     ( CharacterTable( kk[i]), CharacterTable( kk[j]) ) ;
   if not tp = fail then Add( brps, [ IdGroup(kk[i]) , IdGroup(kk[j]) ] );
   fi;
  od;
 od;
 od;
od;
return( brps );
end;
```

We apply the above function to the collections gg of groups of order n (up to isomorphisms) for all $n < 256$ which are not powers of two:

```
gap> brps := [];;
gap> for n in [1..255] do
>      if not Set( Factors(n) ) = [2] then
>        gg := AllSmallGroups(n);
>        Append( brps, braupairli(gg) );
>      fi;
>    od;
gap> time;
333764
```

```
gap> brps;
[  ]
```

The result shows that no Brauer pairs exist among groups of order $< 2^8$ which are not 2-groups. The computation took less than six minutes on an ordinary notebook (of 2009). Looking closer one can see that in the above computation 4344 groups were considered and `TransformingPermutationsCharacterTables` was used 1341 times.

We close by testing the above function on some collections of groups of order 256:

```
gap> gg := List( [1700..1800], m -> SmallGroup( 256, m ) );;
gap> braupairli( gg );
[ [ [ 256, 1741 ], [ 256, 1742 ] ], [ [ 256, 1739 ], [ 256, 1740 ] ],
  [ [ 256, 1736 ], [ 256, 1737 ] ], [ [ 256, 1734 ], [ 256, 1735 ] ] ]
gap> time;
13145
gap> gg := List( [3300..4200], m -> SmallGroup( 256, m ) );;
gap> braupairli( gg );
[ [ [ 256, 3678 ], [ 256, 3679 ] ], [ [ 256, 3379 ], [ 256, 3381 ] ],
  [ [ 256, 3378 ], [ 256, 3380 ] ], [ [ 256, 4156 ], [ 256, 4159 ] ],
  [ [ 256, 4155 ], [ 256, 4158 ] ], [ [ 256, 4154 ], [ 256, 4157 ] ] ]
gap> time;
115484
```

We have found ten Brauer pairs among groups of order 256 which is in accordance with Theorem 2.6.2. It is not feasible running the program on the list of all groups of order 256. So the computation certainly does not show that these are all Brauer pairs among the groups of order 256.